Symmetrised derivatives of the connection

Here we compute, just for a check, the symmetrised derivatives of the connections. These are defined by

$$\Gamma^{d}{}_{a(b,\underline{c})} = \Gamma^{d}{}_{a(b,c_1,c_2,\cdots c_n)} \tag{1}$$

Note that these are *not* the generalised connections. The generalised connections involve $\Gamma^{d}{}_{(ab,\underline{c})}$ and quadratice combinations of lower order generalised connections (see eq (1) of `../genGamma.pdf`). Note that the generalised connections vanish at the origin (unlike the $\Gamma^{d}{}_{a(b,\underline{c})}$).

These results agree with those of Hatzinikitas equation (12) (arXiv:hep-th/0001078).

This code provides an indirect check on our results for the connection. It does not prove that the code `connection.tex` is correct but it does show that our results are consistent with those of Hatzinikitas.

```
{a,b,c,d,e,f,g,h,i,j,k,l,m,n,o,p,q,r,s,t,u,v,w#}::Indices(position=independent).

D{#}::Derivative.
\nabla{#}::Derivative.
\partial{#}::PartialDerivative.

g_{a b}::Metric.
g^{a b}::InverseMetric.
g_{a}^{b}::KroneckerDelta.
g^{a}_{b}::KroneckerDelta.
\delta^{a}_{b}::KroneckerDelta.
\delta_{a}^{b}::KroneckerDelta.

R_{a b c d}::RiemannTensor.
R^{a}_{b c d}::RiemannTensor.
R_{a b c}^{d}::RiemannTensor.

\Gamma^{a}_{b c}::TableauSymmetry(shape={2}, indices={1,2}).

x^{a}::Depends(D{#}).

g_{a b}::Depends(\partial{#}).
R_{a b c d}::Depends(\partial{#}).
R^{a}_{b c d}::Depends(\partial{#}).
\Gamma^{a}_{b c}::Depends(\partial{#}).

R_{a b c d}::Depends(\nabla{#}).
R^{a}_{b c d}::Depends(\nabla{#}).

import cdblib

Gamma = cdblib.get ('Gamma','../connection.json')
```

```
tmp := D_{p}{@(Gamma)}.
distribute      (tmp)
unwrap          (tmp)
product_rule    (tmp)
distribute      (tmp)
substitute      (tmp,$D_{a}{x^{b}}->\delta_{a}^{b}$)
eliminate_kronecker (tmp)
substitute      (tmp,$x^{a}->0$)
sort_product    (tmp)
rename_dummies (tmp)
canonicalise    (tmp)

foo := A^{p} A^{b} @(tmp).
distribute      (foo) # cdb(foo.301,foo)
sort_product    (foo) # cdb(foo.302,foo)
rename_dummies (foo) # cdb(foo.303,foo)
canonicalise    (foo) # cdb(foo.304,foo)

# save the result
dGamma1 := @(foo).    # cdb (dGamma1.000,dGamma1)
```

$$\texttt{foo.301} := \frac{1}{3}A^{p}A^{b}R_{acbp}g^{dc} + \frac{1}{3}A^{p}A^{b}R_{apbc}g^{dc}$$

$$\texttt{foo.302} := \frac{1}{3}A^{b}A^{p}R_{acbp}g^{dc} + \frac{1}{3}A^{b}A^{p}R_{apbc}g^{dc}$$

$$\texttt{foo.303} := \frac{1}{3}A^{b}A^{c}R_{aebc}g^{de} + \frac{1}{3}A^{b}A^{c}R_{acbe}g^{de}$$

$$\texttt{foo.304} := \frac{1}{3}A^{b}A^{c}R_{abce}g^{de}$$

```
tmp := D_{p q}{@(Gamma)}.
distribute      (tmp)
unwrap          (tmp)
product_rule    (tmp)
distribute      (tmp)
substitute      (tmp,$D_{a}{x^{b}}->\delta_{a}^{b}$)
unwrap          (tmp)
product_rule    (tmp)
distribute      (tmp)
substitute      (tmp,$D_{a}{x^{b}}->\delta_{a}^{b}$)
eliminate_kronecker (tmp)
substitute      (tmp,$x^{a}->0$)
sort_product    (tmp)
rename_dummies (tmp)
canonicalise    (tmp)

foo := A^{p} A^{q} A^{b} @(tmp).
distribute      (foo) # cdb(foo.401,foo)
sort_product    (foo) # cdb(foo.402,foo)
rename_dummies (foo) # cdb(foo.403,foo)
canonicalise    (foo) # cdb(foo.404,foo)

# save the result
dGamma2 := @(foo).   # cdb (dGamma2.000,dGamma2)
```

$$\texttt{foo.401} := \frac{1}{12}A^pA^qA^b\nabla_aR_{bqpc}g^{dc} + \frac{1}{12}A^pA^qA^b\nabla_aR_{bpqc}g^{dc} + \frac{1}{6}A^pA^qA^b\nabla_qR_{acbp}g^{dc} + \frac{1}{6}A^pA^qA^b\nabla_pR_{acbq}g^{dc} + \frac{1}{12}A^pA^qA^b\nabla_bR_{aqpc}g^{dc} + \frac{1}{12}A^pA^qA^b\nabla_bR_{apqc}g^{dc} + \frac{1}{6}A^pA^qA^b\nabla_qR_{apbc}g^{dc} + \frac{1}{6}A^pA^qA^b\nabla_pR_{aqbc}g^{dc} + \frac{1}{12}A^pA^qA^b\nabla_cR_{aqbp}g^{dc} + \frac{1}{12}A^pA^qA^b\nabla_cR_{apbq}g^{dc}$$

$$\texttt{foo.402} := \frac{1}{12}A^bA^pA^q\nabla_aR_{bqpc}g^{dc} + \frac{1}{12}A^bA^pA^q\nabla_aR_{bpqc}g^{dc} + \frac{1}{6}A^bA^pA^q\nabla_qR_{acbp}g^{dc} + \frac{1}{6}A^bA^pA^q\nabla_pR_{acbq}g^{dc} + \frac{1}{12}A^bA^pA^q\nabla_bR_{aqpc}g^{dc} + \frac{1}{12}A^bA^pA^q\nabla_bR_{apqc}g^{dc} + \frac{1}{6}A^bA^pA^q\nabla_qR_{apbc}g^{dc} + \frac{1}{6}A^bA^pA^q\nabla_pR_{aqbc}g^{dc} + \frac{1}{12}A^bA^pA^q\nabla_cR_{aqbp}g^{dc} + \frac{1}{12}A^bA^pA^q\nabla_cR_{apbq}g^{dc}$$

$$\texttt{foo.403} := \frac{1}{12}A^bA^cA^e\nabla_aR_{becf}g^{df} + \frac{1}{12}A^bA^cA^e\nabla_aR_{bcef}g^{df} + \frac{1}{6}A^bA^cA^e\nabla_eR_{afbc}g^{df} + \frac{1}{6}A^bA^cA^e\nabla_cR_{afbe}g^{df} + \frac{1}{12}A^bA^cA^e\nabla_bR_{aecf}g^{df} + \frac{1}{12}A^bA^cA^e\nabla_bR_{acef}g^{df} + \frac{1}{6}A^bA^cA^e\nabla_eR_{acbf}g^{df} + \frac{1}{6}A^bA^cA^e\nabla_cR_{aebf}g^{df} + \frac{1}{12}A^bA^cA^e\nabla_fR_{aebc}g^{df} + \frac{1}{12}A^bA^cA^e\nabla_fR_{acbe}g^{df}$$

$$\texttt{foo.404} := \frac{1}{2}A^bA^cA^e\nabla_bR_{acef}g^{df}$$

```
tmp := D_{p q r}{@(Gamma)}.
distribute      (tmp)
unwrap          (tmp)
product_rule    (tmp)
distribute      (tmp)
substitute      (tmp,$D_{a}{x^{b}}->\delta_{a}^{b}$)
unwrap          (tmp)
product_rule    (tmp)
distribute      (tmp)
substitute      (tmp,$D_{a}{x^{b}}->\delta_{a}^{b}$)
unwrap          (tmp)
product_rule    (tmp)
distribute      (tmp)
substitute      (tmp,$D_{a}{x^{b}}->\delta_{a}^{b}$)
eliminate_kronecker (tmp)
substitute      (tmp,$x^{a}->0$)
sort_product    (tmp)
rename_dummies (tmp)
canonicalise    (tmp)

foo := A^{p} A^{q} A^{r} A^{b} @(tmp).
distribute      (foo) # cdb(foo.501,foo)
sort_product    (foo) # cdb(foo.502,foo)
rename_dummies (foo) # cdb(foo.503,foo)
canonicalise    (foo) # cdb(foo.504,foo)

# save the result
dGamma3 := @(foo).    # cdb (dGamma3.000,dGamma3)
```

$$\texttt{foo.504} := \frac{2}{15} A^b A^c A^e A^f R_{abcg} R_{ehfi} g^{dh} g^{gi} + \frac{3}{5} A^b A^c A^e A^f \nabla_{bc} R_{aefg} g^{dg}$$

```
tmp := D_{p q r s}{@(Gamma)}.
distribute      (tmp)
unwrap          (tmp)
product_rule    (tmp)
distribute      (tmp)
substitute      (tmp,$D_{a}{x^{b}}->\delta_{a}^{b}$)
unwrap          (tmp)
product_rule    (tmp)
distribute      (tmp)
substitute      (tmp,$D_{a}{x^{b}}->\delta_{a}^{b}$)
unwrap          (tmp)
product_rule    (tmp)
distribute      (tmp)
substitute      (tmp,$D_{a}{x^{b}}->\delta_{a}^{b}$)
unwrap          (tmp)
product_rule    (tmp)
distribute      (tmp)
substitute      (tmp,$D_{a}{x^{b}}->\delta_{a}^{b}$)
eliminate_kronecker (tmp)
substitute      (tmp,$x^{a}->0$)
sort_product    (tmp)
rename_dummies (tmp)
canonicalise    (tmp)

foo := A^{p} A^{q} A^{r} A^{s} A^{b} @(tmp).
distribute      (foo)
sort_product    (foo)
rename_dummies (foo)
canonicalise    (foo)

# save the result
dGamma4 := @(foo).   # cdb (dGamma4.000,dGamma4)
```

```
tmp := D_{p q r s t}{@(Gamma)}.
distribute      (tmp)
unwrap          (tmp)
product_rule    (tmp)
distribute      (tmp)
substitute      (tmp,$D_{a}{x^{b}}->\delta_{a}^{b}$)
unwrap          (tmp)
product_rule    (tmp)
distribute      (tmp)
substitute      (tmp,$D_{a}{x^{b}}->\delta_{a}^{b}$)
unwrap          (tmp)
product_rule    (tmp)
distribute      (tmp)
substitute      (tmp,$D_{a}{x^{b}}->\delta_{a}^{b}$)
unwrap          (tmp)
product_rule    (tmp)
distribute      (tmp)
substitute      (tmp,$D_{a}{x^{b}}->\delta_{a}^{b}$)
unwrap          (tmp)
product_rule    (tmp)
distribute      (tmp)
substitute      (tmp,$D_{a}{x^{b}}->\delta_{a}^{b}$)
eliminate_kronecker (tmp)
substitute      (tmp,$x^{a}->0$)
sort_product    (tmp)
rename_dummies (tmp)
canonicalise    (tmp)

foo := A^{p} A^{q} A^{r} A^{s} A^{t} A^{b} @(tmp).
distribute      (foo)
sort_product    (foo)
rename_dummies (foo)
canonicalise    (foo)

# save the result
dGamma5 := @(foo).   # cdb (dGamma5.000,dGamma5)
```

Compare these results against those of Hatzinikitas equation (12) (arXiv:hep-th/0001078). Our final $d\Gamma$ is zero because our metric was expanded to order x^5 so the Γ only contain terms to order x^4. Hence the 5-th partial derivatices are zero.

$$A^b A^c \Gamma^d{}_{ab,c} = \frac{1}{3} A^b A^c R_{abce} g^{de} \tag{dGamma1.000}$$

$$A^b A^c A^e \Gamma^d{}_{ab,ce} = \frac{1}{2} A^b A^c A^e \nabla_b R_{acef} g^{df} \tag{dGamma2.000}$$

$$A^b A^c A^e A^f \Gamma^d{}_{ab,cef} = \frac{2}{15} A^b A^c A^e A^f R_{abcg} R_{ehfi} g^{dh} g^{gi} + \frac{3}{5} A^b A^c A^e A^f \nabla_{bc} R_{aefg} g^{dg} \tag{dGamma3.000}$$

$$A^b A^c A^e A^f A^g \Gamma^d{}_{ab,cefg} = \frac{2}{3} A^b A^c A^e A^f A^g R_{bhci} \nabla_e R_{afgj} g^{dh} g^{ij} + \frac{2}{3} A^b A^c A^e A^f A^g \nabla_{bce} R_{afgh} g^{dh} \tag{dGamma4.000}$$

$$A^b A^c A^e A^f A^g A^h \Gamma^d{}_{ab,cefgh} = 0 \tag{dGamma5.000}$$