From one RNC to another

Consider an RNC frame with RNC cooridnates x^a.

In the `geodesic-bvp` code the two point boundary value problem (for the geodesic connecting two points) was solved. There is a bonus in that calculation – it can be trivaly adapted to the case of transforming form one RNC into another.

The starting point is the basic equation for the geodesic connecting P (with coordinaties x^a) to Q (with coordinates $x^a + Dx^a$)

$$x^a(s) = x^a_i + sy^a - \sum_{k=2}^{\infty} \frac{1}{k!}\, \Gamma^a{}_{\underline{b}_k} y^{\underline{b}_k} s^k$$

The affine parameter s varies form 0 (at P) to 1 (at Q).

A new RNC frame, with origin at P, can be defined via the y^a with the coordinates of Q in the new RNC frame defined by y^a (since $s = 1$ at Q). Recall that in an RNC all geodesics through the origin are described by $y^a(s) = sy^a$. Thus the transformation from x^a to y^a satisfies

$$x^a = x^a_i + y^a - \sum_{k=2}^{\infty} \frac{1}{k!}\, \Gamma^a{}_{\underline{b}_k} y^{\underline{b}_k}$$

where the $\Gamma^a{}_{\underline{b}_k}$ are the generalised connections of the x^a frame evaluated at $x^a = 0$. This equation can be inverted to express y^a in terms of x^a. This computation is done in the `geodesic-bvp` code – we only quote the results here (at the end).

The new y^a frame has origin at P. Its coordinate axes are aligned with those (at P) of the origianl RNC frame. To see this just note that $\partial x^a/\partial y^b = \delta^a_b$ at P. Thus the metric at P in the new frame has values $g_{ab}(x)$ (i.e., exactly those of the original RNC frame). Note that this means that the coordinate axes of the new frame are not ncessarily orthogonal.

The calculations in this code are trivial. It uses the y^a found in `geodesic-bvp` as the basis of the transformation from x^a to y^a. Most of the code involves reformatting the y^a.

```
{a,b,c,d,e,f,g,h,i,j,k,l,m,n,o,p,q,r,s,t,u,v,w#}::Indices(position=independent).

\nabla{#}::Derivative.

g_{a b}::Metric.
g^{a b}::InverseMetric.

R_{a b c d}::RiemannTensor.
R^{a}_{b c d}::RiemannTensor.

# Dx{#}::LaTeXForm("{\Dx}").  # LCB: currently causes a bug, it kills ::KeepWeight for Dx

import cdblib

Y5 = cdblib.get ('y5','geodesic-bvp.json')

Y50 = cdblib.get ('y50','geodesic-bvp.json')
Y52 = cdblib.get ('y52','geodesic-bvp.json')
Y53 = cdblib.get ('y53','geodesic-bvp.json')
Y54 = cdblib.get ('y54','geodesic-bvp.json')
Y55 = cdblib.get ('y55','geodesic-bvp.json')

# this copies y5* from geodesic-bvp.json to rnc2rnc.json

cdblib.create ('rnc2rnc.json')

cdblib.put ('rnc2rnc',Y5,'rnc2rnc.json')

cdblib.put ('rnc2rnc0',Y50,'rnc2rnc.json')
cdblib.put ('rnc2rnc2',Y52,'rnc2rnc.json')
cdblib.put ('rnc2rnc3',Y53,'rnc2rnc.json')
cdblib.put ('rnc2rnc4',Y54,'rnc2rnc.json')
cdblib.put ('rnc2rnc5',Y55,'rnc2rnc.json')
```

```
# note: keeping numbering as is (out of order) to ensure R appears before \nabla R etc.
def product_sort (obj):
    substitute (obj,$ x^{a}                             -> A001^{a}               $)
    substitute (obj,$ Dx^{a}                            -> A002^{a}               $)
    substitute (obj,$ g^{a b}                           -> A003^{a b}             $)
    substitute (obj,$ \nabla_{e f g h}{R_{a b c d}}     -> A008_{a b c d e f g h} $)
    substitute (obj,$ \nabla_{e f g}{R_{a b c d}}       -> A007_{a b c d e f g}   $)
    substitute (obj,$ \nabla_{e f}{R_{a b c d}}         -> A006_{a b c d e f}     $)
    substitute (obj,$ \nabla_{e}{R_{a b c d}}           -> A005_{a b c d e}       $)
    substitute (obj,$ R_{a b c d}                       -> A004_{a b c d}         $)
    sort_product   (obj)
    rename_dummies (obj)
    substitute (obj,$ A001^{a}                  -> x^{a}                          $)
    substitute (obj,$ A002^{a}                  -> Dx^{a}                         $)
    substitute (obj,$ A003^{a b}                -> g^{a b}                        $)
    substitute (obj,$ A004_{a b c d}            -> R_{a b c d}                    $)
    substitute (obj,$ A005_{a b c d e}          -> \nabla_{e}{R_{a b c d}}        $)
    substitute (obj,$ A006_{a b c d e f}        -> \nabla_{e f}{R_{a b c d}}      $)
    substitute (obj,$ A007_{a b c d e f g}      -> \nabla_{e f g}{R_{a b c d}}    $)
    substitute (obj,$ A008_{a b c d e f g h}    -> \nabla_{e f g h}{R_{a b c d}} $)

    return obj

def get_xDxterm (obj,n,m):

    x^{a}::Weight(label=numx,value=1).
    Dx^{a}::Weight(label=numDx,value=1).

    tmp := @(obj).
    distribute  (tmp)

    foo = Ex("numx = " + str(n))
    bah = Ex("numDx = " + str(m))
    keep_weight (tmp, foo)
    keep_weight (tmp, bah)

    return tmp
```

```
def reformat (obj,scale):
    foo  = Ex(str(scale))
    bah := @(foo) @(obj).
    distribute     (bah)
    bah = product_sort (bah)
    rename_dummies (bah)
    canonicalise   (bah)
    substitute     (bah,$Dx^{b}->zzz^{b}$)
    factor_out     (bah,$x^{a?},zzz^{b?}$)
    substitute     (bah,$zzz^{b}->Dx^{b}$)
    ans := @(bah) / @(foo).
    return ans

def rescale (obj,scale):
    foo  = Ex(str(scale))
    bah := @(foo) @(obj).
    distribute  (bah)
    substitute  (bah,$Dx^{b}->zzz^{b}$)
    factor_out  (bah,$x^{a?},zzz^{b?}$)
    substitute  (bah,$zzz^{b}->Dx^{b}$)
    return bah

term0 := @(Y50).  # cdb (term0.101,term0)
term2 := @(Y52).  # cdb (term2.101,term2)
term3 := @(Y53).  # cdb (term3.101,term3)
term4 := @(Y54).  # cdb (term4.101,term4)
term5 := @(Y55).  # cdb (term5.101,term5)

term0 = reformat (term0,1)  # cdb (term0.102,term0)
term2 = reformat (term2,1)  # cdb (term2.102,term2)
term3 = reformat (term3,1)  # cdb (term3.102,term3)
term4 = reformat (term4,1)  # cdb (term4.102,term4)
term5 = reformat (term5,1)  # cdb (term5.102,term5)

xDxterm12 = get_xDxterm (term2,1,2)   # cdb(xDxterm12.101,xDxterm12)

xDxterm13 = get_xDxterm (term3,1,3)   # cdb(xDxterm13.101,xDxterm13)
xDxterm22 = get_xDxterm (term3,2,2)   # cdb(xDxterm22.101,xDxterm22)
```

```
xDxterm14 = get_xDxterm (term4,1,4)   # cdb(xDxterm14.101,xDxterm14)
xDxterm23 = get_xDxterm (term4,2,3)   # cdb(xDxterm23.101,xDxterm23)
xDxterm32 = get_xDxterm (term4,3,2)   # cdb(xDxterm32.101,xDxterm32)

xDxterm15 = get_xDxterm (term5,1,5)   # cdb(xDxterm15.101,xDxterm15)
xDxterm24 = get_xDxterm (term5,2,4)   # cdb(xDxterm24.101,xDxterm24)
xDxterm33 = get_xDxterm (term5,3,3)   # cdb(xDxterm33.101,xDxterm33)
xDxterm42 = get_xDxterm (term5,4,2)   # cdb(xDxterm42.101,xDxterm42)


xDxterm12 = rescale ( reformat (xDxterm12,    3),     3 )   # cdb(xDxterm12.102,xDxterm12)

xDxterm13 = rescale ( reformat (xDxterm13,   12),   -12 )   # cdb(xDxterm13.102,xDxterm13)
xDxterm22 = rescale ( reformat (xDxterm22,   24),   -24 )   # cdb(xDxterm22.102,xDxterm22)

xDxterm14 = rescale ( reformat (xDxterm14,  180),  -180 )   # cdb(xDxterm14.102,xDxterm14)
xDxterm23 = rescale ( reformat (xDxterm23,  720),  -720 )   # cdb(xDxterm23.102,xDxterm23)
xDxterm32 = rescale ( reformat (xDxterm32,  720),  -720 )   # cdb(xDxterm32.102,xDxterm32)

xDxterm15 = rescale ( reformat (xDxterm15,  360),  -360 )   # cdb(xDxterm15.102,xDxterm15)
xDxterm24 = rescale ( reformat (xDxterm24, 2160), -2160 )   # cdb(xDxterm24.102,xDxterm24)
xDxterm33 = rescale ( reformat (xDxterm33, 1080), -1080 )   # cdb(xDxterm33.102,xDxterm33)
xDxterm42 = rescale ( reformat (xDxterm42,  360),  -360 )   # cdb(xDxterm42.102,xDxterm42)

checkpoint.append (term0)
checkpoint.append (term2)
checkpoint.append (term3)
checkpoint.append (term4)
checkpoint.append (term5)
```

Tranformation between two RNC frames

$$y^a = \overset{0}{y}{}^a + \overset{2}{y}{}^a + \overset{3}{y}{}^a + \overset{4}{y}{}^a + \overset{5}{y}{}^a + \mathcal{O}\left(\epsilon^6\right)$$

$$\overset{0}{y}{}^a = Dx^a$$

$$\overset{2}{y}{}^a = -\frac{1}{3}x^b Dx^c Dx^d g^{ae} R_{bcde}$$

$$\overset{3}{y}{}^a = x^b x^c Dx^d Dx^e \left(-\frac{1}{12}g^{af}\nabla_d R_{becf} - \frac{1}{6}g^{af}\nabla_b R_{cdef} + \frac{1}{24}g^{af}\nabla_f R_{bdce}\right) - \frac{1}{12}x^b Dx^c Dx^d Dx^e g^{af}\nabla_c R_{bdef}$$

$$\begin{aligned}\overset{4}{y}{}^a = {} & x^b x^c Dx^d Dx^e Dx^f \left(-\frac{2}{45}g^{ag}g^{hi}R_{bdeh}R_{cfgi} + \frac{1}{45}g^{ag}g^{hi}R_{bdeh}R_{cifg} - \frac{1}{45}g^{ag}g^{hi}R_{bdeh}R_{cgfi} + \frac{1}{45}g^{ag}g^{hi}R_{bdch}R_{egfi} - \frac{1}{60}g^{ag}\nabla_{de}R_{bfcg} - \frac{1}{40}g^{ag}\nabla_{db}R_{cefg}\right. \\ & \left. - \frac{1}{40}g^{ag}\nabla_{bd}R_{cefg} + \frac{1}{240}g^{ag}\nabla_{gd}R_{becf} + \frac{1}{240}g^{ag}\nabla_{dg}R_{becf}\right) \\ & + x^b x^c x^d Dx^e Dx^f \left(-\frac{4}{45}g^{ag}g^{hi}R_{befh}R_{cgdi} + \frac{2}{45}g^{ag}g^{hi}R_{bech}R_{difg} + \frac{1}{45}g^{ag}g^{hi}R_{bech}R_{dgfi} - \frac{1}{40}g^{ag}\nabla_{eb}R_{cfdg} - \frac{1}{40}g^{ag}\nabla_{be}R_{cfdg} - \frac{1}{20}g^{ag}\nabla_{bc}R_{defg}\right. \\ & \left. - \frac{1}{45}g^{ag}g^{hi}R_{bech}R_{dfgi} + \frac{1}{80}g^{ag}\nabla_{gb}R_{cedf} + \frac{1}{80}g^{ag}\nabla_{bg}R_{cedf}\right) + x^b Dx^c Dx^d Dx^e Dx^f \left(-\frac{1}{45}g^{ag}g^{hi}R_{bcdh}R_{egfi} - \frac{1}{60}g^{ag}\nabla_{cd}R_{befg}\right)\end{aligned}$$

$$
\begin{aligned}
\overset{5}{y}{}^a = x^b x^c x^d Dx^e Dx^f Dx^g \Bigg(& -\frac{7}{540} g^{ah} g^{ij} R_{behi} \nabla_f R_{cgdj} - \frac{1}{45} g^{ah} g^{ij} R_{behi} \nabla_c R_{dfgj} + \frac{1}{216} g^{ah} g^{ij} R_{behi} \nabla_j R_{cfdg} + \frac{1}{90} g^{ah} g^{ij} R_{bieh} \nabla_f R_{cgdj} \\
& + \frac{1}{90} g^{ah} g^{ij} R_{bieh} \nabla_c R_{dfgj} - \frac{1}{540} g^{ah} g^{ij} R_{befi} \nabla_g R_{chdj} + \frac{1}{108} g^{ah} g^{ij} R_{befi} \nabla_j R_{cgdh} - \frac{1}{45} g^{ah} g^{ij} R_{befi} \nabla_c R_{dghj} + \frac{1}{90} g^{ah} g^{ij} R_{befi} \nabla_c R_{djgh} \\
& - \frac{7}{540} g^{ah} g^{ij} R_{befi} \nabla_h R_{cgdj} - \frac{1}{90} g^{ah} g^{ij} R_{befi} \nabla_c R_{dhgj} - \frac{1}{540} g^{ah} g^{ij} R_{bhei} \nabla_f R_{cgdj} - \frac{11}{540} g^{ah} g^{ij} R_{bhci} \nabla_e R_{dfgj} - \frac{1}{90} g^{ah} g^{ij} R_{bhei} \nabla_c R_{dfgj} \\
& + \frac{1}{216} g^{ah} g^{ij} R_{bhei} \nabla_j R_{cfdg} + \frac{1}{90} g^{ah} g^{ij} R_{ehfi} \nabla_b R_{cgdj} + \frac{1}{90} g^{ah} g^{ij} R_{beci} \nabla_f R_{djgh} + \frac{1}{135} g^{ah} g^{ij} R_{beci} \nabla_f R_{dhgj} + \frac{1}{90} g^{ah} g^{ij} R_{beci} \nabla_d R_{fhgj} \\
& - \frac{1}{270} g^{ah} \nabla_{efb} R_{cgdh} - \frac{1}{270} g^{ah} \nabla_{ebf} R_{cgdh} - \frac{1}{180} g^{ah} \nabla_{ebc} R_{dfgh} - \frac{1}{270} g^{ah} \nabla_{bef} R_{cgdh} - \frac{1}{180} g^{ah} \nabla_{bec} R_{dfgh} - \frac{1}{180} g^{ah} \nabla_{bce} R_{dfgh} - \frac{1}{270} g^{ah} g^{ij} R_{beci} \nabla_h R_{dfgj} \\
& - \frac{1}{270} g^{ah} g^{ij} R_{beci} \nabla_f R_{dghj} + \frac{1}{1080} g^{ah} \nabla_{heb} R_{cfdg} + \frac{1}{1080} g^{ah} \nabla_{hbe} R_{cfdg} + \frac{1}{1080} g^{ah} \nabla_{ehb} R_{cfdg} + \frac{1}{1080} g^{ah} \nabla_{bhe} R_{cfdg} + \frac{1}{1080} g^{ah} \nabla_{ebh} R_{cfdg} \\
& + \frac{1}{1080} g^{ah} \nabla_{beh} R_{cfdg} \Bigg) + x^b x^c Dx^d Dx^e Dx^f Dx^g \Bigg(-\frac{17}{1080} g^{ah} g^{ij} R_{bdhi} \nabla_e R_{cfgj} + \frac{1}{135} g^{ah} g^{ij} R_{bidh} \nabla_e R_{cfgj} - \frac{7}{1080} g^{ah} g^{ij} R_{bdei} \nabla_f R_{cghj} \\
& - \frac{1}{540} g^{ah} g^{ij} R_{bdei} \nabla_f R_{cjgh} + \frac{1}{108} g^{ah} g^{ij} R_{bdei} \nabla_j R_{cfgh} - \frac{1}{120} g^{ah} g^{ij} R_{bdei} \nabla_f R_{chgj} - \frac{1}{90} g^{ah} g^{ij} R_{bdei} \nabla_c R_{fhgj} - \frac{1}{540} g^{ah} g^{ij} R_{dhei} \nabla_f R_{bgcj} \\
& - \frac{1}{120} g^{ah} g^{ij} R_{bhdi} \nabla_e R_{cfgj} - \frac{1}{90} g^{ah} g^{ij} R_{dhei} \nabla_b R_{cfgj} + \frac{1}{216} g^{ah} g^{ij} R_{dhei} \nabla_j R_{bfcg} + \frac{1}{135} g^{ah} g^{ij} R_{bdci} \nabla_e R_{fhgj} - \frac{1}{360} g^{ah} \nabla_{def} R_{bgch} - \frac{1}{270} g^{ah} \nabla_{deb} R_{cfgh} \\
& - \frac{1}{270} g^{ah} \nabla_{dbe} R_{cfgh} - \frac{1}{270} g^{ah} \nabla_{bde} R_{cfgh} + \frac{1}{360} g^{ah} g^{ij} R_{bdei} \nabla_h R_{cfgj} + \frac{1}{2160} g^{ah} \nabla_{hde} R_{bfcg} + \frac{1}{2160} g^{ah} \nabla_{dhe} R_{bfcg} + \frac{1}{2160} g^{ah} \nabla_{deh} R_{bfcg} \Bigg) \\
& + x^b x^c x^d x^e Dx^f Dx^g \Bigg(-\frac{2}{45} g^{ah} g^{ij} R_{bfgi} \nabla_c R_{dhej} - \frac{1}{60} g^{ah} g^{ij} R_{bhci} \nabla_f R_{dgej} - \frac{2}{45} g^{ah} g^{ij} R_{bhci} \nabla_d R_{efgj} + \frac{1}{72} g^{ah} g^{ij} R_{bhci} \nabla_j R_{dfeg} \\
& + \frac{1}{45} g^{ah} g^{ij} R_{bifh} \nabla_c R_{dgej} + \frac{1}{90} g^{ah} g^{ij} R_{bhfi} \nabla_c R_{dgej} + \frac{1}{90} g^{ah} g^{ij} R_{bfci} \nabla_g R_{dhej} + \frac{1}{45} g^{ah} g^{ij} R_{bfci} \nabla_d R_{ejgh} + \frac{1}{90} g^{ah} g^{ij} R_{bfci} \nabla_d R_{ehgj} \\
& - \frac{1}{180} g^{ah} \nabla_{fbc} R_{dgeh} - \frac{1}{180} g^{ah} \nabla_{bfc} R_{dgeh} - \frac{1}{180} g^{ah} \nabla_{bcf} R_{dgeh} - \frac{1}{90} g^{ah} \nabla_{bcd} R_{efgh} - \frac{1}{90} g^{ah} g^{ij} R_{bfhi} \nabla_c R_{dgej} - \frac{1}{90} g^{ah} g^{ij} R_{bfci} \nabla_h R_{dgej} \\
& - \frac{1}{90} g^{ah} g^{ij} R_{bfci} \nabla_d R_{eghj} + \frac{1}{360} g^{ah} \nabla_{hbc} R_{dfeg} + \frac{1}{360} g^{ah} \nabla_{bhc} R_{dfeg} + \frac{1}{360} g^{ah} \nabla_{bch} R_{dfeg} \Bigg) \\
& + x^b Dx^c Dx^d Dx^e Dx^f Dx^g \Bigg(-\frac{1}{120} g^{ah} g^{ij} R_{bcdi} \nabla_e R_{fhgj} - \frac{1}{120} g^{ah} g^{ij} R_{chdi} \nabla_e R_{bfgj} - \frac{1}{360} g^{ah} \nabla_{cde} R_{bfgh} \Bigg)
\end{aligned}
$$

Tranformation between two RNC frames

Same as before but with an improved format (maybe) for the expressions.

$$y^a = \overset{0}{y}{}^a + \overset{2}{y}{}^a + \overset{3}{y}{}^a + \overset{4}{y}{}^a + \overset{5}{y}{}^a + \mathcal{O}\left(\epsilon^6\right) \tag{1}$$

$$\overset{0}{y}{}^a = Dx^a \tag{2a}$$

$$\overset{2}{y}{}^a = \overset{2}{y}{}^a_1 \tag{3a}$$

$$3\overset{2}{y}{}^a_1 = -x^b Dx^c Dx^d g^{ae} R_{bcde} \tag{3b}$$

$$\overset{3}{y}{}^a = \overset{3}{y}{}^a_1 + \overset{3}{y}{}^a_2 \tag{4a}$$

$$-12\overset{3}{y}{}^a_1 = x^b Dx^c Dx^d Dx^e g^{af} \nabla_c R_{bdef} \tag{4b}$$

$$-24\overset{3}{y}{}^a_2 = x^b x^c Dx^d Dx^e \left(2g^{af}\nabla_d R_{becf} + 4g^{af}\nabla_b R_{cdef} - g^{af}\nabla_f R_{bdce}\right) \tag{4c}$$

$$\overset{4}{y}{}^a = \overset{4}{y}{}^a_1 + \overset{4}{y}{}^a_2 + \overset{4}{y}{}^a_3 \tag{5a}$$

$$-180\overset{4}{y}{}^a_1 = x^b Dx^c Dx^d Dx^e Dx^f \left(4g^{ag}g^{hi}R_{bcdh}R_{egfi} + 3g^{ag}\nabla_{cd}R_{befg}\right) \tag{5b}$$

$$\begin{aligned}-720\overset{4}{y}{}^a_2 = x^b x^c Dx^d Dx^e Dx^f \big(&32g^{ag}g^{hi}R_{bdeh}R_{cfgi} - 16g^{ag}g^{hi}R_{bdeh}R_{cifg} + 16g^{ag}g^{hi}R_{bdeh}R_{cgfi} - 16g^{ag}g^{hi}R_{bdch}R_{egfi} + 12g^{ag}\nabla_{de}R_{bfcg} \\ &+ 18g^{ag}\nabla_{db}R_{cefg} + 18g^{ag}\nabla_{bd}R_{cefg} - 3g^{ag}\nabla_{gd}R_{becf} - 3g^{ag}\nabla_{dg}R_{becf}\big)\end{aligned} \tag{5c}$$

$$\begin{aligned}-720\overset{4}{y}{}^a_3 = x^b x^c x^d Dx^e Dx^f \big(&64g^{ag}g^{hi}R_{befh}R_{cgdi} - 32g^{ag}g^{hi}R_{bech}R_{difg} - 16g^{ag}g^{hi}R_{bech}R_{dgfi} + 18g^{ag}\nabla_{eb}R_{cfdg} + 18g^{ag}\nabla_{be}R_{cfdg} \\ &+ 36g^{ag}\nabla_{bc}R_{defg} + 16g^{ag}g^{hi}R_{bech}R_{dfgi} - 9g^{ag}\nabla_{gb}R_{cedf} - 9g^{ag}\nabla_{bg}R_{cedf}\big)\end{aligned} \tag{5d}$$

$$\overset{5}{y}{}^a = \overset{5}{y}{}^a_1 + \overset{5}{y}{}^a_2 + \overset{5}{y}{}^a_3 + \overset{5}{y}{}^a_4 \tag{6a}$$

$$-360\overset{5}{y}{}^a_1 = x^b Dx^c Dx^d Dx^e Dx^f Dx^g \left(3g^{ah}g^{ij}R_{bcdi}\nabla_e R_{fhgj} + 3g^{ah}g^{ij}R_{chdi}\nabla_e R_{bfgj} + g^{ah}\nabla_{cde}R_{bfgh}\right) \tag{6b}$$

$$\begin{aligned}-2160\overset{5}{y}{}^a_2 = x^b x^c Dx^d Dx^e Dx^f Dx^g \big(&34g^{ah}g^{ij}R_{bdhi}\nabla_e R_{cfgj} - 16g^{ah}g^{ij}R_{bidh}\nabla_e R_{cfgj} + 14g^{ah}g^{ij}R_{bdei}\nabla_f R_{cghj} + 4g^{ah}g^{ij}R_{bdei}\nabla_f R_{cjgh}\\ &- 20g^{ah}g^{ij}R_{bdei}\nabla_j R_{cfgh} + 18g^{ah}g^{ij}R_{bdei}\nabla_f R_{chgj} + 24g^{ah}g^{ij}R_{bdei}\nabla_c R_{fhgj} + 4g^{ah}g^{ij}R_{dhei}\nabla_f R_{bgcj} + 18g^{ah}g^{ij}R_{bhdi}\nabla_e R_{cfgj}\\ &+ 24g^{ah}g^{ij}R_{dhei}\nabla_b R_{cfgj} - 10g^{ah}g^{ij}R_{dhei}\nabla_j R_{bfcg} - 16g^{ah}g^{ij}R_{bdci}\nabla_e R_{fhgj} + 6g^{ah}\nabla_{def}R_{bgch} + 8g^{ah}\nabla_{deb}R_{cfgh} + 8g^{ah}\nabla_{dbe}R_{cfgh}\\ &+ 8g^{ah}\nabla_{bde}R_{cfgh} - 6g^{ah}g^{ij}R_{bdei}\nabla_h R_{cfgj} - g^{ah}\nabla_{hde}R_{bfcg} - g^{ah}\nabla_{dhe}R_{bfcg} - g^{ah}\nabla_{deh}R_{bfcg}\big)\end{aligned} \tag{6c}$$

$$\begin{aligned}-1080\overset{5}{y}{}^a_3 = x^b x^c x^d Dx^e Dx^f Dx^g \big(&14g^{ah}g^{ij}R_{behi}\nabla_f R_{cgdj} + 24g^{ah}g^{ij}R_{behi}\nabla_c R_{dfgj} - 5g^{ah}g^{ij}R_{behi}\nabla_j R_{cfdg} - 12g^{ah}g^{ij}R_{bieh}\nabla_f R_{cgdj}\\ &- 12g^{ah}g^{ij}R_{bieh}\nabla_c R_{dfgj} + 2g^{ah}g^{ij}R_{befi}\nabla_g R_{chdj} - 10g^{ah}g^{ij}R_{befi}\nabla_j R_{cgdh} + 24g^{ah}g^{ij}R_{befi}\nabla_c R_{dghj} - 12g^{ah}g^{ij}R_{befi}\nabla_c R_{djgh}\\ &+ 14g^{ah}g^{ij}R_{befi}\nabla_h R_{cgdj} + 12g^{ah}g^{ij}R_{befi}\nabla_c R_{dhgj} + 2g^{ah}g^{ij}R_{bhei}\nabla_f R_{cgdj} + 22g^{ah}g^{ij}R_{bhci}\nabla_e R_{dfgj} + 12g^{ah}g^{ij}R_{bhei}\nabla_c R_{dfgj}\\ &- 5g^{ah}g^{ij}R_{bhei}\nabla_j R_{cfdg} - 12g^{ah}g^{ij}R_{ehfi}\nabla_b R_{cgdj} - 12g^{ah}g^{ij}R_{beci}\nabla_f R_{djgh} - 8g^{ah}g^{ij}R_{beci}\nabla_f R_{dhgj} - 12g^{ah}g^{ij}R_{beci}\nabla_d R_{fhgj}\\ &+ 4g^{ah}\nabla_{efb}R_{cgdh} + 4g^{ah}\nabla_{ebf}R_{cgdh} + 6g^{ah}\nabla_{ebc}R_{dfgh} + 4g^{ah}\nabla_{bef}R_{cgdh} + 6g^{ah}\nabla_{bec}R_{dfgh} + 6g^{ah}\nabla_{bce}R_{dfgh} + 4g^{ah}g^{ij}R_{beci}\nabla_h R_{dfgj}\\ &+ 4g^{ah}g^{ij}R_{beci}\nabla_f R_{dghj} - g^{ah}\nabla_{heb}R_{cfdg} - g^{ah}\nabla_{hbe}R_{cfdg} - g^{ah}\nabla_{ehb}R_{cfdg} - g^{ah}\nabla_{bhe}R_{cfdg} - g^{ah}\nabla_{ebh}R_{cfdg} - g^{ah}\nabla_{beh}R_{cfdg}\big)\end{aligned} \tag{6d}$$

$$\begin{aligned}-360\overset{5}{y}{}^a_4 = x^b x^c x^d x^e Dx^f Dx^g \big(&16g^{ah}g^{ij}R_{bfgi}\nabla_c R_{dhej} + 6g^{ah}g^{ij}R_{bhci}\nabla_f R_{dgej} + 16g^{ah}g^{ij}R_{bhci}\nabla_d R_{efgj} - 5g^{ah}g^{ij}R_{bhci}\nabla_j R_{dfeg}\\ &- 8g^{ah}g^{ij}R_{bifh}\nabla_c R_{dgej} - 4g^{ah}g^{ij}R_{bhfi}\nabla_c R_{dgej} - 4g^{ah}g^{ij}R_{bfci}\nabla_g R_{dhej} - 8g^{ah}g^{ij}R_{bfci}\nabla_d R_{ejgh} - 4g^{ah}g^{ij}R_{bfci}\nabla_d R_{ehgj}\\ &+ 2g^{ah}\nabla_{fbc}R_{dgeh} + 2g^{ah}\nabla_{bfc}R_{dgeh} + 2g^{ah}\nabla_{bcf}R_{dgeh} + 4g^{ah}\nabla_{bcd}R_{efgh} + 4g^{ah}g^{ij}R_{bfhi}\nabla_c R_{dgej} + 4g^{ah}g^{ij}R_{bfci}\nabla_h R_{dgej}\\ &+ 4g^{ah}g^{ij}R_{bfci}\nabla_d R_{eghj} - g^{ah}\nabla_{hbc}R_{dfeg} - g^{ah}\nabla_{bhc}R_{dfeg} - g^{ah}\nabla_{bch}R_{dfeg}\big)\end{aligned} \tag{6e}$$